ARPA ORDER NO. 139-1

MEMORANDUM
RM-5546-ARPA
JULY 1968

AD 672588

# ALTERNATING DIRECTION EXPLICIT-IMPLICIT COMPUTATIONAL METHOD APPLIED TO LOW-FREQUENCY UNDERWATER SOUND PROPAGATION

M. C. Smith, R. H. Mayall and S. V. Huber



PREPARED FOR:

ADVANCED RESEARCH PROJECTS AGENCY

The RAND Corporation
SANTA MONICA • CALIFORNIA



30

ARPA ORDER NO. 189-1

MEMORANDUM

RM-5546-ARPA

JULY 1968

# ALTERNATING DIRECTION EXPLICIT-IMPLICIT COMPUTATIONAL METHOD APPLIED TO LOW-FREQUENCY UNDERWATER SOUND PROPAGATION

M. C. Smith, R. H. Mayall and S. V. Huber

This research is supported by the Advanced Research Projects Agency under Contract No. DAHC15 67 C 0141. RAND Memoranda are subject to critical review procedures at the research department and corporate levels. Views and conclusions expressed herein are nevertheless the primary responsibility of the author, and should not be interpreted as representing the official opinion or policy of ARPA or of The RAND Corporation.

DISTRIBUTION STATEMENT

This document has been approved for public release and sale; its distribution is unlimited.

The RAND Corporation

1700 MAIN ST. • SANTA MONICA • CALIFORNIA • 90406

## PREFACE

In sonar evaluation studies there are at present three approximate methods of determining sound propagation: ray theory, mode theory and semi-empirical formulation. Under the usual environmental conditions, one or a combination of two of these methods will produce satisfactory prediction of intensity versus range. But under extreme environmental circumstances, favorable or unfavorable, range prediction is not reliable with any combination of these methods. Another limitation of the existing methods arises with the increasing reliance on signal processing. Static methods of range prediction are becoming less useful even under favorable circumstances.

This Memorandum, part of a larger study conducted for the Advanced Research Projects Agency, presents a technique which may prove useful under difficult environmental conditions and in situations in which a dynamic solution is required. This latter capability should be of special interest to those who are concerned with sonar development in particular, and with acoustic propagation in general.

## SUMMARY

A computer solution of the wave-difference equations is found by using an interlacing explicit-implicit scheme. In the computation, the entire two-dimensional field is found as a function of time.

The examples considered involve propagation in a homogeneous shallow-water channel where the effect of superposition of discrete spectra produces characteristic modal patterns. The spatial sound pressure fluctuations are represented as plot-density variations along the two-dimensional channel. Examples of discrete propagated modes as well as evanescent modes are presented. The size of the field that can be presented is limited by the size, accuracy, and speed of the computer.

# CONTENTS

## LIST OF FIGURES

BLANK PAGE

# I. INTRODUCTION

In this study the difference equations representing wave motion are solved by an interlacing explicit-implicit procedure. This scheme in its present form was applied with remarkable success by J. J. Leendertse to the solution of long-period ocean wave motion.(1) Reference 1 demonstrated that this method is capable of producing accurate stable solutions of the nonlinear wave equation without the usual restricting assumptions. The present study is concerned with the two-dimensional linearized undamped wave equation.

## II. DIFFERENCE EQUATIONS

The basic first-order undamped differential equations which describe the sound field are[2]

$$\rho \frac{\partial U}{\partial t} + \frac{\partial (C^2 \xi)}{\partial x} = 0 \qquad (1)$$

$$\rho \frac{\partial W}{\partial t} + \frac{\partial (C^2 \xi)}{\partial z} = 0 \qquad (2)$$

$$\frac{\partial \xi}{\partial t} + \rho \frac{\partial U}{\partial x} + \rho \frac{\partial W}{\partial z} = 0 \qquad (3)$$

where the quantities $\xi$, U, W represent the linearized departures in the variables, density, and two components of particle velocity, respectively. In this Memorandum, the sound velocity C is considered only as a function of the depth z; the density $\rho$ is a constant. Both of these quantities could be functions of spatial as well as of time coordinates.

The difference equations which represent the differential Eqs. (1), (2) and (3) could be formed in a variety of ways. The scheme chosen here uses a two-interlacing grid system in which the density point is surrounded by two stream points, as indicated in Fig. 1. Position in the grid, formed by these points, is indicated by two subscripts. The first refers to the row and the second to the column.

$$U'_{n,m} = U_{n,m} - \frac{\tau}{\ell \rho} C_n^2 (\xi'_{n,m+1} - \xi'_{n,m}) \qquad (4)$$

$$W'_{n,m} = W_{n,m} - \frac{\tau}{\ell \rho} (C_{n+1}^2 \xi_{n+1,m} - C_n^2 \xi_{n,m}) \qquad (5)$$




Fig. 1—Scheme for the boundaries and the grid location of excess velocity and density in the channel

$$\xi'_{n,m} = \xi_{n,m} - \frac{\rho\tau}{\ell}\left(U'_{n,m} - U'_{n,m-1}\right) - \frac{\rho\tau}{\ell}\left(W_{n,m} - W_{n-1,m}\right) \tag{6}$$

The primed superscripts designate the quantity at $t + \tau$.*

---

*Note another equally valid formulation:

$$U'_{n,m} = U_{n,m} - \frac{\tau}{\ell\rho} C_n^2 \left(\xi_{n,m+1} - \xi_{n,m}\right)$$

$$W'_{n,m} = W_{n,m} - \frac{\tau}{\ell\rho} \left(C_{n+1}^2 \xi'_{n+1,m} - C_n^2 \xi'_{n,m}\right)$$

and

$$\xi'_{n,m} = \xi_{n,m} - \frac{\rho\tau}{\ell}\left(U_{n,m} - U_{n,m-1}\right) - \frac{\rho\tau}{\ell}\left(W'_{n,m} - W'_{n-1,m}\right)$$

This formulation is used in the second half-time step when the alternate solution direction is used.

## III. THE ALTERNATING DIRECTION IMPLICIT-EXPLICIT METHOD OF SOLUTION

During each of two successive time increments $\tau$, a different operation is performed. Equations (1) and (3) are first solved implicitly; Eq. (2) is then solved explicitly in the first time interval. In the second time interval, Eqs. (2) and (3) are solved implicitly and Eq. (1) explicitly. These two operations comprise a one-time step of $2\tau$. From a knowledge of the boundary conditions, the two groups of implicit equations are solved by eliminating the unknowns.

The alternating direction method can be applied directly to Eqs. (4) through (6) if the equations are put in the following form:

First half-time step: implicit solution over the row n

$$- \frac{\tau}{\ell} \rho \, U'_{n,m-1} + \xi'_{n,m} + \frac{\tau}{\ell} \rho \, U'_{n,m} = A_{n,m} \tag{7}$$

$$- \frac{\tau}{\rho \ell} C_n^2 \, \xi'_{n,m} + U'_{n,m} + \frac{\tau}{\rho \ell} C_n^2 \, \xi'_{n,m+1} = B_{n,m} \tag{8}$$

where

$$A_{n,m} = \xi_{n,m} + \frac{\tau}{\ell} \rho \, (W_{n-1,m} - W_{n,m})$$

$$B_{n,m} = U_{n,m}$$

First half-time step: explicit solution over the column m

$$W'_{n,m} = W_{n,m} - \frac{\tau}{\rho} \left\{ \frac{C_{n+1}^2 \, \xi_{n+1,m} - C_n^2 \, \xi_{n,m}}{\ell} \right\} \tag{9}$$

Second half-time step: implicit solution over the column m

$$-\frac{\tau}{\ell}\rho W''_{n-1,m} + \xi''_{n,m} + \frac{\tau}{\ell}\rho W''_{n,m} = a_{n,m} \tag{10}$$

$$-\frac{\tau}{\rho}\left(\frac{C^2_n}{\ell}\xi''_{n,m}\right) + W''_{n,m} + \frac{\tau}{\rho}\left(\frac{C^2_{n+1}}{\ell}\xi''_{n+1,m}\right) = b_{n,m} \tag{11}$$

where

$$a_{n,m} = \xi'_{n,m} - \frac{\tau}{\ell}\rho\left(U'_{n,m} - U'_{n,m-1}\right)$$

$$b_{n,m} = W'_{n,m}$$

Second half-time step: explicit solution over row n

$$U''_{n,m} = U'_{n,m} - \frac{\tau}{\rho}\left\{\frac{C^2_n}{\ell}(\xi'_{n,m+1} - \xi'_{n,m})\right\} \tag{12}$$

During the first half-time step t to t + $\tau$, Eqs. (7) and (8) are solved implicitly, while Eq. (9) is solved explicitly. The implicit formulas for each n during the first half-time step can be written in general terms

$$\xi'_{n,m} = - P_{n,m} U'_{n,m} + Q_{n,m} \tag{13}$$

$$U'_{n,m-1} = - R_{n,m-1}\xi'_{n,m} + S_{n,m-1} \tag{14}$$

where

$$P_{n,m} = \frac{\frac{\tau}{\ell}\rho}{1 + \frac{\tau}{\ell}\rho R_{n,m-1}}$$

$$Q_{n,m} = \frac{A_{n,m} + \frac{\tau}{\ell}\rho\, S_{n,m-1}}{1 + \frac{\tau}{\ell}\rho\, R_{n,m-1}}$$

$$R_{n,m} = \frac{\frac{\tau}{\rho\ell} C_n^2}{1 + \frac{\tau}{\rho\ell} C_n^2 P_{n,m}} \qquad \text{if } m \neq 1$$

$$= 0 \qquad \text{if } m = 1$$

$$S_{n,m} = \frac{B_{n,m} + \frac{\tau}{\rho\ell} C_n^2 Q_{n,m}}{1 + \frac{\tau}{\rho\ell} C_n^2 P_{n,m}} \qquad \text{if } m \neq 1$$

$$= U'_{n,1} \qquad \text{if } m = 1$$

$U'_{n,1}$ represents an external driving function at the zero position; it is zero otherwise. In the following time step $t + \tau$ to $t + 2\tau$, Eqs. (10) and (11) are solved implicitly, while Eq. (12) is solved explicitly. The generalized form of the equations is

$$\xi''_{n,m} = - p_{n,m} W''_{n,m} + q_{n,m} \tag{15}$$

$$W''_{n-1,m} = - r_{n-1,m}\, \xi''_{n,m} + s_{n-1,m} \tag{16}$$

where

$$p_{n,m} = \frac{\frac{\tau}{\ell}\rho}{1 + \frac{\tau}{\ell}\rho\, r_{n-1,m}}$$

$$q_{n,m} = \frac{a_{n,m} + \frac{\tau}{\ell}\rho\, s_{n-1,m}}{1 + \frac{\tau}{\ell}\rho\, r_{n-1,m}}$$

$$r_{n,m} = \frac{\frac{\tau}{\rho}\frac{C^2_{n+1}}{\ell}}{1 + \frac{\tau}{\rho}\frac{C^2_n}{\ell} p_{n,m}} \qquad \text{if } n \neq 1$$

$$= 0 \qquad \text{if } n = 1$$

$$s_{n,m} = \frac{b_{n,m} + \frac{\tau}{\rho}\frac{C^2_n}{\ell} q_{n,m}}{1 + \frac{\tau}{\rho}\frac{C^2_n}{\ell} p_{n,m}} \qquad \text{if } n \neq 1$$

$$= 0 \qquad \text{if } n = 1$$

Equations (13) through (16) are in the form used in the numerical computation.

## IV. ORDER OF SOLUTION

During the first half-time step the coefficients $P_{n,m}$, $Q_{n,m}$, etc. are evaluated in succession over each row beginning with $m = 1$ and extending to the far boundary at $m = m_{max}$. At the far end of the channel, beyond the area where the wave has disturbed the channel, $U'_{n,m_{max}} = 0$, Eq. (13) becomes

$$\xi'_{n,m_{max}} = Q_{n,m_{max}}$$

The solution for Eqs. (13) and (14) begins at $m = m_{max}$ and proceeds to $m = 2$ over each row, beginning with $n = 2$, as follows:

$$\xi'_{n,m_{max}} = + Q_{n,m_{max}}$$

$$U'_{n,m_{max}-1} = - R_{n,m_{max}} \xi'_{n,m_{max}} + S_{n,m_{max}-1}$$

$$\xi'_{n,m_{max}-1} = - P_{n,m_{max}-1} U'_{n,m_{max}-1} + Q_{n,m_{max}-1}$$

$$\vdots \qquad \vdots \qquad \vdots$$

$$U'_{n,2} = - R_{n,2} \xi'_{n,3} + S_{n,2}$$

$$\xi'_{n,2} = - P_{n,2} U'_{n,2} + Q_{n,2}$$

The explicit Eq. (9) is then solved over the columns m beginning with $n = 2$.

In the following half-time step $p_{n,m}$, $q_{n,m}$, etc. are evaluated in succession over each column beginning with n = 2. The implicit Eqs. (15) and (16) for $\xi_{n,m}$, $W_{n,m}$ can then be solved in the same way as the first half-time step, except that the boundary conditions now depend on the surface and bottom of the channel. The water-air interface cannot sustain any change of pressure. The boundary conditions at the free surface of the channel $n = n_{max}$ can be satisfied if $\xi''_{n_{max},m} = 0$. Equation (16) becomes

$$W''_{n_{max}-1,m} = + s_{n_{max}-1,m}$$

Beginning with this equation, at the top of the channel the solution proceeds to the bottom of the channel in a procedure similar to that used to compute the first half-time step. For the rigid boundary at the bottom of the channel (Fig. 1), the velocity in the vertical direction is $W''_{1,m} = 0$. Then Eq. (15) can be written

$$\xi''_{2,m} = - p_{2,m} W''_{2,m} + q_{2,m}$$

## V. DISCUSSION OF THE EXAMPLES

The interlaced implicit-explicit method was programmed in FORTRAN for RAND's 7040-7044 computer for a rectangular channel. Various grid sizes, time increments and frequencies were tried. The numerical stability of the calculations and the basis for grid and time step size have been discussed in Ref. 2. For the examples presented, a 20 cps source of the form

$$A \sin (\omega t + \varphi)$$

was applied at one end of the channel. In order to reduce the transient disturbances, a zero phase angle was used. Peak amplitudes of the steady-state solution of excess density at each data point of the channel were retained and plotted on the S-C 4060 microfilm recorder. A density plot representative of the peak density amplitude was constructed by making the size of the character (X in this case) represent a density amplitude range. In Figs. 2 and 3 the darker regions represent less sonified regions, while the white regions represent those which are highly sonified.* These plots represent only a portion of the steady-state field.

Examples of patterns of locked and evanescent modes are shown in Figs. 2 and 3. As indicated in Fig. 2(a), the source was placed in the center of the end of the channel. This resulted in the energizing of modes 1 and 3. In Fig. 2(b), the source was placed near a nodal point of mode 3; hence mode 3 was suppressed to some extent and energy was

---

*As programmed, the maximum peak amplitude is divided into five equal increments. The first increment (a blank) represents the excess density in the range 1.0A to 0.8A, and the four successive divisions represent decrements in amplitude with the largest X designating the range 0.2A to 0.0.




Fig. 2—Peak amplitude of the excess density in a channel
237.5 ft deep and 1550 ft long
(white regions are highly sonified)




Fig. 3—Peak amplitude of the excess density in a channel 237.5 ft deep and 1550 ft long (white regions are highly sonified)

directed to modes 1 and 2. These two figures represent the channel driven at a frequency below mode 3 cutoff, as the attenuation in the x-direction demonstrates.

In Fig. 3 one boundary condition was changed to represent a free surface, as described in the preceding section. All other parameters are the same, the source is in the middle of the channel, and modes 1 and 2 are propagated while mode 3 and higher modes are evanescent. These results are very much like those of the optical fringe analog of sound propagation of Ref. 3.

It is possible to compare these results with ray theory by verifying that the horizontal pattern repetition distance $\Lambda$ given by ray theory is the same as the $\Lambda$ indicated in Fig. 3. From ray theory[3]

$$\Lambda = \frac{8H^2}{\lambda(n^2 - m^2)} \tag{18}$$

where n and m are the mode numbers, H the distance between two density release surfaces, and $\lambda$ the wavelength. The phase change at the rigid surface is equivalent to a reflection at a free surface lying at a distance

$$H' = \frac{1}{\sin\theta}\frac{\lambda}{4} \tag{19}$$

below the rigid surface. Here $\theta$ is the angle that the ray makes with the boundary for in-phase regions to appear. From Fig. 3 this angle is approximately $20^{\circ}$. The enhanced channel depth is then

$$H = H' + H_1 \tag{20}$$

From the parameters assumed in Figs. (2) and (3)

sound velocity $V$ = 5000 ft/sec

wavelength $\lambda$ = 250 ft

channel depth $H_1$ = 237.5 ft

grid size $\ell$ = 12.5 ft

time step size $\tau$ = 0.00125 sec

the horizontal repetition length calculated from Eqs. (18), (19) and (20) is approximately 739 ft, while the measured distance from Fig. 3 is approximately 12.5 x 57.0 ft = 712.5 ft. The difference between the ray theory calculation and the calculation in this work probably can be attributed to quantization.

## VI. CONCLUSIONS

It is important to recognize the distinctive features and the inherent limitations of the implicit-explicit method compared with the methods now in use. It is difficult to infer all of the features and limitations of this method from these simple examples, but the following are evident:

### FEATURES

1. The entire field is found as a function of time in the calculation.
2. Other than the grid size, no restrictions are placed on the velocity field and boundary conditions.
3. The parameters describing the field can vary in time.

### LIMITATIONS

1. The field size is limited by the accuracy and the storage capacity of the computer.
2. The computation time is a function of the field size and the computer fast-storage capacity. For example, one hour of RAND 7040-7044 computer time and most of the fast-storage capacity were consumed in producing each solution as shown in Figs. 2 and 3.

It is generally accepted that ray tracing, normal mode, or semi-empirical method, or some combination of them, is adequate for most naval applications, provided that the environmental parameters and sonar system parameters are available and are used correctly. However, in conditions which are either exceptionally good or exceptionally bad, all three methods fail in reliable range prediction.[4] The implicit-explicit method has a potential which is not available with current methods for range prediction under the conditions of a variable local environment.

# Appendix

## COMPUTER PROGRAM

SØUNDW is a main routine written in FØRTRAN IV for use on the IBM-7044, but it also may be used on the various machines capable of using the FØRTRAN language.

Input, read in by FØRMAT (I5), consists of:

1. Length of grid - MDIM
2. Width of grid - NMAX
3. Expanding boundary for M - MMAX
4. Maximum number of time steps - MAXST
5. Initial time step number - NST
6. Frequency - F
7. Length of grid interval - AL
8. Velocity - VEL
9. Time step size - AT
10. Rho - RHØ
11. Gamma - GAMMA
12. Amplitude - AMPU

Output is stored on FØRTRAN I/Ø unit 9

1. SEP, SEPP - density
2. UP, UPP - X-component of velocity
3. WP, WPP - Z-component of velocity

where P refers to the first half-time step, and PP refers to the second half-time step.

```
$IBFTC SOUNDW   REF
C
C      SIMULATES SOUND PROPAGATION UNDERWATER
C
C                                       **********
C                                       * SOUNDW *
C                                       **********
C
C      M.C. SMITH
C      THE RAND CORP.
C      SANTA MONICA,CALIF.
C
C      WRITTEN BY R.H. MAYALL, JULY,1967
C      REVISED BY S.V. HUBER, DECEMBER,1967
C
       DIMENSION POIT(20,124)
       DIMENSION U(20,222),SE(20,222),W(20,222),WP(20,222)
       DIMENSION SEP(20,222),SEPP(20,222),UP(20,222),UPP(20,222),
      1  WPP(20,222)
       EQUIVALENCE (SE(1,1),SEP(1,1),SEPP(1,1))
       EQUIVALENCE (U(1,1),UP(1,1))
       EQUIVALENCE (W(1,1),UPP(1,1))
       EQUIVALENCE (WP(1,1),WPP(1,1))
       DIMENSION A(222),B(222),P(222),Q(222),R(222),S(222)
       DIMENSION XDAV(222),X1DAV(222)
       DIMENSION CSQ(21)
       DIMENSION DES(6),DES1(2)
       DIMENSION  NO(222)
       DATA DES /3H  U,4H  UP,4H  SE,4H SEP,3H  W,4H  WP/
       DATA DES1 /5HFIRST,6HSECOND/
       READ(5,404) MDIM
       READ(5,404) NMAX
       READ(5,404) MMAX
       READ(5,404) MAXST
       READ(5,404) NST
       READ(5,406) F
       READ(5,406) AL
       READ(5,406) VEL
       READ(5,406) AT
       READ(5,406) RHO
       READ(5,406) GAMMA
       READ(5,406) AMPU
 404   FORMAT(I5)
 406   FORMAT(F10.4)
       DO 10 I=1,MDIM
 10    NO(I) = I
C
C**    MMAX SHOULD BE EVEN INITIALLY.
       NMAX1 = NMAX-1
       MMAX1 = MMAX-1
       AL = AL*30.5
       CF = .975914506
       VEL = VEL*30.5
       C1 = AT/AL
       C2 = C1/AL
C
C      INITIAL CONDITIONS
       DO 20 M=1,MDIM
        A(M) = 0.
        B(M) = 0.
```

```
       P(M) = 0.
       Q(M) = 0.
       R(M) = 0.
       S(M) = 0.
      DO 20 N=1,NMAX
      U(N,M) = 0.
      SE(N,M) = 0.0
      W(N,M) = 0.
 20   WP(N,M. = 0.
      PI = 3.1415926
      FPI = 4.0*PI*F*AT
      TROL = C1*RHO
      PL2 = C1/AL/RHO
      PLGAM = PL2*GAMMA
      DO 30 N=2,NMAX
       CSQ(N) = C1*VEL**2/RHO
 30   CONTINUE
C
C     XDAV,X1DAV USED ONLY FOR A CYLINDRICAL COORDINATE SYSTEM
      DO 40 I=2,MDIM
      XM = AL*FLOAT(I-1)
      XMM= AL*FLOAT(I-2)
      XDAV(I) = 2.*XM/(XM+XMM)*TROL
      X1DAV(I) = 2.*XMM/(XM+XMM)*TROL
 40   CONTINUE
C
C     FIRST HALF-TIME STEP CALCULATIONS BEGIN
 50   CONTINUE
C
C     EXPLICIT CALCULATION OF WP
      XNST = NST
      DO 70  M=2,MMAX
      MM = M-1
      MMM = M+1
      DO 60  N=2,NMAX1
      NN = N-1
      NNN = N+1
      HOLD1 = W(N,M) - CSQ(NNN)*SE(NNN,M) + CSQ(N)*SE(N,M)
      WP(N,M) = HOLD1 + PLGAM*(W(NNN,M) - 2.0*W(N,M) + W(NN,M))
 60   CONTINUE
      WP(NMAX,M) = 0.0
 70   CONTINUE
C
C     R(N,1),S(N,1) DEFINE BDRY. CONDS. FOR UP
      R(1) = 0.0
      L = MMAX
C
C     IMPLICIT CALCULATION OF P,Q,R,S
      DO 100 N=2,NMAX
      S(1) = 0.0
C
C     POINT SOURCE
      IF(N.EQ.14)   S(1) = AMPU*SIN(FPI*XNST)
      NN = N-1
      DO 80 M=2,L
      MM = M-1
      MMM = M+1
      A(M) = SE(N,M) + TROL*(W(NN,M) - W(N,M))
      PDEM = 1.0 + TROL*R(MM)
      P(M) = TROL/PDEM
```

```
      Q(M) =(A(M) +    TROL*S(MM))/PDEM
      IF(M.EQ.MMAX) GO TO 80
      B(M) = U(N,M) +     PLGAM*(U(N,MMM) -2.0*U(N,M) + U(N,MM))
      RNUM = CSQ(N)
      RDEM = 1.0 + RNUM*P(M)
      R(M) = RNUM/RDEM
      S(M) = (B(M) + RNUM*Q(M))/RDEM
 80   CONTINUE
      UP(N,L)= 0.0
      M=L
C
C      IMPLICIT CALCULATION OF SEP,UP
      DO 90  J=2,MMAX
      MM = M-1
      SEP(N,M) =-P(M)*UP(N,M)+Q(M)
      UP(N,MM) =-R(MM)*SEP(N,M)+S(MM)
 90   M = M-1
 100  CONTINUE
      MX = MMAX
      ASSIGN 110 TO KK
      GO TO 190
C
C      SECOND HALF-TIME STEP CALCULATIONS BEGIN
 110  CONTINUE
C
C      EXPLICIT CALCULATION OF UPP
      DO 130 N=2,NMAX
      NN = N-1
      NNN = N+1
      DO 120 M=2,MMAX1
      MM = M-1
      MMM = M+1
      UPP(N,M)=UP(N,M) -      CSQ(N)*(SEP(N,MMM) - SEP(N,M)) + PLGAM*
     1(UP(N,MMM)-2.0*UP(N,M) + UP(N,MM))
 120  CONTINUE
 130  CONTINUE
C
C      BOUNDARY CONDITIONS ON UPP
      DO 140 N=1,NMAX
       UPP(N,1) = 0.0
 140   UPP(N,MMAX) = 0.0
      UPP(14,1) = AMPU*SIN(FPI*(XNST+.5))
C
C      IMPLICIT CALCULATION OF P,Q,R,S
      DO 170 M=2,MMAX
C
C      R(1,M),S(1,M) DEFINE BDRY. CONDS. FOR WPP
      R( 1 ) = 0.0
      S( 1 ) = 0.0
      MM = M-1
      MMM = M+1
      DO 150 N=2,NMAX
      NN = N-1
      NNN = N+1
      A(N) = SEP(N,M) -    TROL*(UP(N,M)-UP(N,MM))
      PDE = 1.0 +TROL*R(NN)
      P(N) =TROL/PDE
      Q(N) =(A(N) +TROL*S(NN))/PDE
      IF(N.EQ.NMAX) GO TO 150
      B(N)=WP(N,M)+   PLGAM*(WP(NNN,M)-2.*WP(N,M)+WP(NN,M))
```

```
      RDE = 1.0 + CSQ(N)*P(N)
      R(N) = CSQ(NNN)/RDE
      S(N) =(B(N) + Q(N)*CSQ(N))/RDE
  150 CONTINUE
      CON1 = CSQ(NMAX)
      N= NMAX
C
C     BOUNDARY CONDITIONS ON WPP
      WPP(NMAX,M) = 0.0
C
C     IMPLICIT CALCULATION OF SEPP,WPP
      DO 160 J=2,NMAX
      NN = N-1
      SEPP(N,M) = -P(N)*WPP(N,M) + Q(N)
      WPP(NN,M) = -R(NN)*SEPP(N,M) + S(NN)
  160 N = N-1
  170 CONTINUE
      DO 180 J=1,MMAX
      DO 180 I=1,NMAX
        U(I,J) = UPP(I,J)
  180   W(I,J) = WPP(I,J)
      ASSIGN 200 TO KK
  190 CONTINUE
      DO 2 N=1,20
      DO 3 M=1,124
      IF(SE(N,M).GT.POIT(N,M))  POIT(N,M) = SE(N,M)
    3 CONTINUE
    2 CONTINUE
      GO TO KK,(110,200)
  200 CONTINUE
      NST = NST+1
C
C     EXPANDING BOUNDARY
      IF(MMAX.LT.MDIM) MMAX = MMAX+2
      MMAX1 = MMAX-1
      IF(NST.GT.MAXST) GO TO 210
      GO TO 50
  210 CONTINUE
      WRITE(9) POIT
      DO 13  M=1,124
      PRINT 14, (POIT(N,M), N=1,20)
   13 CONTINUE
      STOP
   14 FORMAT (1H ,10E13.5)
  300 FORMAT(1H ,I3,2X,11E11.4)
  301 FORMAT(1H1,4X,4HTHE ,A6,9HVALUES AT,1X,A6,17H HALF OF TIMESTEP,I5)
  400 FORMAT(1H ,3X,I3)
  402 FORMAT(1H0,6HNST = ,I4,23HALL VALUES ARE AVERAGED)
      END
$ENTRY          SOUNDW
  222                  LENGTH OF GRID
   20                  WIDTH  OF GRID
   60                  EXPANDING BOUNDRY FOR M
  350                  MAX. NO. OF TIMESTEPS
    1                  INITIAL TIMESTEP NO.
       20.0            FREQUENCY
       12.5            LENGTH OF GRID INTERVAL
     5000.0            VELOCITY
     .00125            TIMESTEP SIZE
      1.025            RHO
```

```
     .0114          GAMMA
      6.0           AMPLITUDE
$IBSYS
$CLOSE          S.SU06,REMOVE

$IBFTC PLOTP
      DIMENSION AMODES(200)
      DIMENSION POIT(20,124)
      DIMENSION SIZE(4),X(2500),Y(2500)
      DATA SIZE/.75,1.0,1.25,1.5/
      READ (9) POIT
      CALL MODESG(AMODES,0)
      CALL SETSMG (AMODES,84,6HXXXXXX)
      CALL SETSMG (AMODES,94,1.)
      CALL OBJCTG(AMODES,0.,0.,1.333333,.3)
      CALL SUBJEG (AMODES,0.0,0.0,124.,20.)
      CALL GRIDG (AMODES,1.0,1.0,0,0)
      DO 20 I=1,4
      IX = 0
      DO 30 K=1,20
      DO 40 L=1,124
      GO TC (60,70,80,90), I
   60 IF(POIT(K,L).LT. .50E-5.AND.POIT(K,L).GE. .40E-5)  GO TO 50
      GO TO 40
   70 IF(POIT(K,L).LT. .40E-5.AND.POIT(K,L).GE. .30E-5)  GO TO 50
      GO TO 40
   80 IF(POIT(K,L).LT. .30E-5.AND.POIT(K,L).GE. .20E-5)  GO TO 50
      GO TO 40
   90 IF(POIT(K,L).LT. .20E-5.AND.POIT(K,L).GE. .00E-1)  GO TO 50
      GO TO 40
   50 IX = IX+1
      X(IX) = L
      Y(IX) = K
   40 CONTINUE
   30 CONTINUE
      CALL SETSMG(AMODES,53,SIZE(I))
      CALL POINTG (AMODES,IX,X,Y)
   20 CONTINUE
      CALL PICTRG(AMODES,1,0,1)
      CALL EXITG (AMODES)
      CALL EXIT
      END
```

## REFERENCES


1. Leendertse, J. J., Aspects of a Computational Model for Long-Period Water-Wave Propagation, The RAND Corporation, RM-5294-PR, May 1967.

2. Gray, Dwight E. (ed.), American Institute of Physics Handbook, 2d ed., McGraw-Hill Book Company, Inc., New York, 1963, pp. 3-35.

3. Weston, D. E., "A Moiré Fringe Analog of Sound Propagation in Shallow Water," J. Acoust. Soc. Am., Vol. 32, June 1960, pp. 647-654.

4. ASW Advisory Committee, National Security Industrial Association, Report of the Detection and Classification Subcommittee AD HOC Study on Acoustic Propagation Analysis Techniques, Serial No. 11-65, undated.

DOCUMENT CONTROL DATA

| | |
|---|---|
| 1 ORIGINATING ACTIVITY<br>THE RAND CORPORATION | 2a. REPORT SECURITY CLASSIFICATION<br>UNCLASSIFIED<br>2b. GROUP |
| 3. REPORT TITLE<br>ALTERNATING DIRECTION EXPLICIT-IMPLICIT COMPUTATIONAL METHOD APPLIED TO LOW-FREQUENCY UNDERWATER SOUND PROPAGATION | |
| 4 AUTHOR(S) (Last name, first name, initial)<br>Smith, M. C., R. H. Mayall and S. V. Huber | |

| | | |
|---|---|---|
| 5. REPORT DATE<br>July 1968 | 6a. TOTAL No. OF PAGES<br>32 | 6b. No. OF REFS.<br>4 |
| 7. CONTRACT OR GRANT No.<br>DAHC15 67 C 0141 | 8. ORIGINATOR'S REPORT No.<br>RM-5546-ARPA | |

| | |
|---|---|
| 9a. AVAILABILITY / LIMITATION NOTICES<br>DDC-1 | 9b. SPONSORING AGENCY<br>Advanced Research Projects Agency |

10. ABSTRACT

A new sonar interpretation technique for range determination under conditions in which the usual approximate methods are inadequate. A computer solution of the two-dimensional undamped linearized wave equations is found by using an interlacing explicit-implicit scheme. In the computation, the entire two-dimensional field is found as a function of time. The examples considered involve wave propagation in a homogeneous shallow-water channel where the effect of superposition of discrete spectra produces characteristic modal patterns. The spatial sound pressure fluctuations are represented as plot-density variations along the two-dimensional channel. Examples of discrete propagated modes as well as evanescent modes are presented. The size of the field that can be presented is limited by the size, accuracy, and speed of the computer. The method is programmed in FORTRAN IV for use on the IBM-7040-7044 computer with the S-C 4060 microfilm recorder. The program is appended.

11. KEY WORDS

Sonar
Sound
Wave propagation
Computer graphics
Numerical methods and processes